PLASTIC DESIGN OF TWO HINGED LATTICED CIRCULAR ARCHES

By

MERVIN L. SNOWDEN

Bachelor of Architectural Engineering

Oklahoma State University

Stillwater, Oklahoma

1964

Submitted to the faculty of the Graduate College of
the Oklahoma State University
in partial fulfillment of the requirements
for the degree of
MASTER OF ARCHITECTURAL ENGINEERING
May, 1966

PLASTIC DESIGN OF TWO HINGED
LATTICED CIRCULAR ARCHES

Thesis Approved:

Thesis Adviser

Dean of the Graduate College

ACKNOWLEDGEMENTS

I wish to express my sincere appreciation to the following individuals whose help and guidance have made my graduate studies possible.

Professor Louis O. Bass, my adviser, whose technical advice and assistance has made this study possible.

Professor F. Cuthbert Salmon, who aided in obtaining a graduate assistantship, which made it possible for me to continue my education.

My wife, whose encouragement and self-sacrifice have been invaluable to me in my graduate study. Also whose typing skill is evidenced on the pages of this study.

My parents, whose assistance and encouragement has made my higher education possible.

My brother, who encouraged me to seek a higher degree and whose assistance has been helpful in my years of study.

TABLE OF CONTENTS

LIST OF TABLES

LIST OF FIGURES

LIST OF SYMBOLS

EQUATIONS	COMPUTER PROGRAM	
d	D	Depth of the arch
d_p		Depth from centerline of arch to the pressure line
e	E	Vertical distance from arch center of curvature to base line of arch
FB_I		Force in bottom chord member
FP_I		Force in diagonal web member
FT_I		Force in top chord member
FV_I		Force in radial web member
H_A	HA	Horizontal reaction at left support
H_B	HB	Horizontal reaction at right support
K	K	Joint on top chord about which plastic rotation occurs
KCL	KCL	Joint at center of top chord
L	L	Joint on bottom chord about which plastic rotation occurs
MB_I	XMB(I)	Moment about a bottom chord joint
MT_I	XMT(I)	Moment about a top chord joint
N	N	Number of panels

EQUATIONS	COMPUTER PROGRAM	
P	P(I)	Concentrated load
PD_I		Concentrated dead load
PDR_I		Concentrated drift load
PL_I		Concentrated live load
PM	PM	Plastic moment
$\frac{PM}{d}$		Force in yielded chord member
R	R	Radius of curvature of arch centerline
S	HS	Half span of arch
SP	SP	Span of arch
T	T	Height of centerline of arch
THR_I	THR(I)	Thrust in panel I
V_A	VA	Vertical reaction at left support
V_B	VB	Vertical reaction at right support
V_I	SH(I)	Shear in panel I
W_D	WD	Dead load
W_{DR}	WDR	Drift load
W_L	WL	Live load
W_W		Wind load
XB_I	XB(I)	Horizontal distance from left support to joint I on bottom chord

EQUATIONS	COMPUTER PROGRAM	
XT_I	XT(I)	Horizontal distance from left support to joint I on top chord
YB_I	YB(I)	Vertical distance from base line of arch to joint I on bottom chord
YT_I	YT(I)	Vertical distance from base line of arch to joint I on top chord
α	ALFA	Angle measured from horizontal to a radius line to the left support
β		Angle measured from horizontal to a radius line to the right support
$d\varphi$	DPHI	Angle between radius lines drawn to adjacent joints
φ	PHI	Central angle of arch
ρ	RHO	Angle measured from horizontal to a radius line to the joint being considered
θ		Angle between diagonal web member and bottom chord member

CHAPTER I

INTRODUCTION

The method of plastic design has principally been applied to the design of statically indeterminate structures of mild steel with rolled sections. These structures carry their loads primarily by the resistance of their members to bending action. The plastic methods depend on the ability of the rolled steel sections to develop plastic hinges which can undergo a large rotation while the bending moment remains constant. The use of plastic design simplifies the design calculations because the plastic methods render the statically indeterminate structure statically determinate at collapse. Plastic design usually yields a saving of material.

Plastic design has been extended to the design of latticed portal frames by Fisher, Heyman and Jaeger (4). The result of their study led to a savings of material as compared to an elastic design by reducing the size of members required in the latticed rafter. They also noted that the plastic design calculations were much simpler than the calculations required for the elastic solution. Their study included a full scale test of the structure to destruction. The mode of collapse and the failure load that they predicted was verified by the test.

This thesis theoretically applies to the plastic design of latticed arches. In latticed structures, the plastic hinge is formed by indefinite axial deformation of the member under a constant axial force. Tension members of mild steel exhibit this property, provided that the strains are not too great, but the behavior of compression members is complicated by the buckling phenomena. The behavior of compression members at critical load is discussed by Van Den Broek (6). Because of the complications present in columns, this thesis will assume that the structure is designed such that no compression member will reach its ultimate load before the structure as a whole collapses as a mechanism.

CHAPTER II

THE ANALYSIS

The Structure

The structure that is analyzed in this thesis is shown in Figure 1. The connections are all assumed to be frictionless pins. The joints are located on a circular arc and the members are straight members between the joints.



Figure 1. The Structure to be Analyzed

The Loads

Dead load, live load or snow load, drifted snow load and wind load are the loads that are considered in this thesis. Beedle, et al. (2) recommends the following combinations of the above loads to determine the critical design forces.

I. 1.85 (Dead + Snow)

II. 1.40 (Dead + Wind)

III. 1.40 (Dead + Wind + Snow)

IV. 1.40 (Dead + Wind + Partial Snow)

The partial snow load in combination IV. will be considered to be drifted snow load in this thesis. The wind load used



Figure 2. The Uniform Loads Applied to the Center Line of the Arch

is the loading recommended by the ASCE Sub-Committee 31. The uniform loads as applied to the center line of the structure are shown in Figure 2. Cornforth (3) gives wind load factors that may be used to determine W_1, W_2, and W_3 in Figure 2D. In this thesis, the uniform loads, as shown in Figure 2 are resolved into concentrated loads and applied at the joints along the top chord of the latticed arch as shown in Figure 3.



Figure 3. The Concentrated Loads Applied to the Latticed Arch

The Pressure Line

The pressure line is a line, broken or curved, which represents the successive resultants in their correct position along a structure, of all the forces acting on the structure. For uniform live load across the arch, the pressure line takes the form of a parabola. As the arch varies from semi-circle, or a pitch of one to two, to a smaller rise to span ratio, the circular segment approaches the pressure line. For unsymmetrical loads, such as drift load, the pressure line is farther away from the arch center line. This concept is illustrated in Figure 4. The pressure line also



Figure 4. Pressure Lines for Live and Drift Loads

represents the shape of the moment diagram. If the moment diagram is plotted perpendicular, to scale, to the arch center line, the moment value divided by the perpendicular distance to the arch center line will yield the thrust, or axial force, present in the arch at that point.

By use of Cornforth's (3) computer program for plastic design of two hinged circular arches, the moments and thrusts were calculated for varying pitches and loadings. From this data, the distance to the pressure line can be calculated. The distance from the center line of the arch to the pressure line at points of plastic hinges is shown in Table 1. The

TABLE I

DISTANCE FROM CENTER LINE OF ARCH TO THE PRESSURE LINE

Loading	Pitch	Plastic Moment PM(Ft-Kips)	Maximum Thrust T(Kips)	Distance to Pressure Line d_p (Ft)
$1.85(W_D+W_L)$	1/8	112.17	12.28	0.11
$1.85(W_D+W_L+P)$	1/8	112.07	16.20	0.14
$1.40(W_D+W_W)$	1/8	6.08	38.34	6.31
$1.40(W_D+W_W+W_L)$	1/8	58.56	32.68	0.56
$1.40(W_D+W_W+W_{DR})$	1/8	18.02	26.64	1.49
$1.85(W_D+W_L)$	1/4	68.49	50.08	0.73
$1.40(W_D+W_W)$	1/4	3.39	42.99	12.68
$1.40(W_D+W_W+W_L)$	1/4	34.07	37.63	1.10
$1.40(W_D+W_W+W_{DR})$	1/4	9.37	18.03	1.92
$1.85(W_D+W_L)$	1/2	54.42	225.31	4.14
$1.40(W_D+W_W)$	1/2	3.31	75.86	22.92
$1.40(W_D+W_W+W_L)$	1/2	29.15	161.91	5.55
$1.40(W_D+W_W+W_{DR})$	1/2	13.06	71.05	5.44

data used to arrive at the values in Table 1 is shown below:

Arches at 12'-0" o. c.

Span = 100'-0"

Dead Load		= 200 lbs/ft.
Live Load	= 30 lbs/sq.ft. x 12 ft.	= 360 lbs/ft.
Drift Load	= 15 lbs/sq.ft. x 12 ft.	= 180 lbs/ft.
Wind Load	= 20 lbs/sq.ft. x 12 ft.	= 240 lbs/ft.
Concentrated Load		= 1000 lbs.

The drift load is applied to the right half of the arch. The wind load is acting from the left and the concentrated load is applied at 25 ft. from the left support. Table 1 indicates that the distance to the pressure line is dependent upon the rise to span ratio of the arch and the loading condition.

Assuming linear stress distribution, Figure 5 shows the state of stress on a rectangular cross section as the distance to the pressure line, d_p, varies from $d_p > d/2$ to $d_p = 0$. For



Figure 5. Stress Distribution on Rectangular Cross Section With Varying Distances to the Pressure Line

$d_p > d/2$, as shown in Figure 5A, the stress varies from compression on one extreme fiber to tension on the other extreme fiber. For $d_p = d/2$, as shown in Figure 5B, the stress varies from compression on one extreme fiber to zero stress on the other extreme fiber. For $d_p < d/2$, as shown in Figure 5C, the entire cross section is in compression. When $d_p = 0$, there is uniform compression across the cross section as shown in Figure 5D.

Therefore, to produce tension on one member of a latticed section, the distance to the pressure line must be greater than the half depth of the section. This condition is necessary to produce the plastic hinge.

The force in the compression member may be greater than the force in the tension member. Also, the compression member is limited because of buckling. Therefore, the compression chord of the latticed section will be required to be a larger size than the tension chord.

Collapse Condition

As the load is increased on the structure, yielding or continuous axial elongation will occur in some member of the structure. The stresses are redistributed and the load can be increased until yielding occurs in another member of the structure. This procedure can repeat until enough members have yielded to produce failure of the structure as a whole, or collapse of the structure. The necessary conditions for collapse of a two hinged latticed arch are shown in Figure 6. Figure 6A shows the condition necessary for the collapse



A. Collapse Condition for Symmetrical Live or Dead Loading



B. Collapse Condition for Unsymmetrical Loading

Figure 6. The Collapse Condition for Symmetrical and Unsymmetrical Loading

of the arch with a uniform dead or live load across the entire arch. Four members must yield before collapse occurs with this type loading on a two hinged latticed arch. Figure 6B shows the collapse condition for an unsymmetrical load. Only two members must yield to produce collapse for an unsymmetrical loading condition on a two hinged latticed arch. In both cases, there is a rotation around a joint on the top chord and a rotation around at least one joint on the bottom chord.



Figure 7. The Structure and Loads

Geometry and Determination of Concentrated Loads

Given the values of SP, T, N, d, and the uniform loads as shown in Figure 7, the following equations can be derived.

Given:

SP = Span of the arch

T = Rise of the center line of the arch

N = Number of panels

d = Depth of the arch

Then:

$$S = \frac{SP}{2}$$

$$R = \text{center line radius} = \frac{T^2 + S^2}{2T}$$

$$e = R - T$$

$$\varphi = 2.0 \; TAN^{-1}\left(\frac{S}{e}\right)$$

$$d\varphi = \frac{\varphi}{N}$$

$$\alpha = TAN^{-1}\left(\frac{e}{S}\right)$$

$$\beta = \alpha + \varphi$$

ρ varies from α to β

$$XB_I = S - \left(R - \frac{d}{2}\right)\cos\rho$$

$$XT_I = S - \left(R + \frac{d}{2}\right)\cos\rho$$

$$YB_I = \left(R - \frac{d}{2} \right) \sin \rho - e$$

$$YT_I = \left(R + \frac{d}{2} \right) \sin \rho - e$$

PL_I = concentrated load due to W_L =

$$= \frac{W_L}{2} \left(XT_{I+1} - XT_{I-1} \right)$$

except at joints 1 and NN,

$$PL_I = \frac{W_L XT_2}{2}$$

PDR_I = concentrated load due to W_{DR} =

$$= \frac{W_{DR}}{2} \left(XT_{I+1} - XT_{I-1} \right)$$

except at joints KCL and NN,

$$PDR_{KCL} = \frac{W_{DR} XT_{KCL+1}}{2}$$

$$PDR_{NN} = \frac{W_{DR} XT_2}{2}$$

PD_I = concentrated load due to W_D = $R \; d\varphi \, W_D$

except at joints 1 and NN,

$$PD_I = \frac{R \; d\varphi \, W_D}{2}$$

Solution for Unknown Reactions

There are four unknown reactions, V_A, V_B, H_A, and H_B. The vertical reactions can be found by the use of statics equations and a relationship between the horizontal reactions can be found by the use of the third statics equation.

$\Sigma M = 0$, and $\Sigma F_Y = 0$ yields:

$$V_A = f(W)$$

$$V_B = f(W)$$

$\Sigma F_X = 0$ yields:

$$H_A = f(H_B)$$

One compatability condition is required for the solution of the horizontal reactions. Continuous axial deformation of a member in the top chord and a member in the bottom chord or rotation about a joint in the top chord and rotation about a joint in the bottom chord gives the required compatability condition. This assumption yields two free body diagrams as shown in Figure 8. A location for joints K and L is assumed and all of the forces in the chord members are found. The forces in the yielded members opposite the joints K and L will be equal and should be the maximum tensile forces existant in the chord members. If this condition is not found, a new location for joints K and L is assumed. This procedure is repeated until the above condition is satisfied. The force in the yielded members is given the notation of PM/d. The force in the yielded members times the depth of the arch is equal to the plastic moment, PM. The joint on the top chord is given the notation of joint K. The joint on the bottom chord is called joint L. For live and dead loads, joint K will be at the center line of the structure as shown in Figure 6A. Figure 6B shows that joint K will be on the right side of the arch for drift loads on the right side of the arch.



Figure 8. Free Body Diagrams at Plastic Hinges

From Figure 8:

$$\Sigma M_K = 0$$

$$\left(V_A - P_1 \right) XT_K - H_A YT_K - P_2 \left(XT_K - XT_2 \right) - \ldots$$

$$\ldots - P_{K-1} \left(XT_K - XT_{K-1} \right) - \left(\frac{PM}{d} \right) d = 0 \qquad \text{or,}$$

<u>EQ. 1</u>:

$$PM = \left(V_A - P_1 \right) XT_K - H_A YT_K - P_2 \left(XT_K - XT_2 \right) - \ldots$$

$$\ldots - P_{K-1} \left(XT_K - XT_{K-1} \right)$$

And,

$$\Sigma M_L = 0$$

$$\left(V_A - P_1 \right) XB_L - H_A YB_L - P_2 \left(XB_L - XT_2 \right) - \ldots$$

$$\ldots - P_L \left(XB_L - XT_L \right) + \left(\frac{PM}{d} \right) d = 0 \qquad \text{or,}$$

<u>EQ. 2</u>:

$$H_A = \frac{PM + \left(V_A - P_1 \right) XB_L - P_2 \left(XB_L - XT_2 \right) - \ldots - P_L \left(XB_L - XT_L \right)}{YB_L}$$

Let

$$R_A = V_A - P_1$$

Substituting Equation 2 into Equation 1 yields:

$$PM = \frac{R_A\left(XT_K YB_L - XB_L YT_K\right) + \left[P_2\left(XB_L - XT_2\right) + \ \ldots \ + P_L\left(XB_L - XT_L\right)\right] YT_K}{YB_L + YT_K} +$$

$$+ \frac{\left[-P_2\left(XT_K - XT_2\right) - \ \ldots \ - P_{K-1}\left(XT_K - XT_{K-1}\right)\right] YB_L}{YB_L + YT_K}$$

H_A can now be found by using Equation 2. Next, the forces in all the members of the arch can be found.

Member Forces

To determine the forces in the members, moments, shears, and thrusts must be calculated across the arch. The forces in the chords can be found by finding the moments at all the joints and dividing these moment values by the depth of the arch. The free body diagrams are shown in Figure 9. The sign convention used is that positive moment yields tension in the chord member.



Figure 9. Free Body Diagrams for Member Forces

For a bottom joint I:

$$\Sigma M_I = 0$$

Yields,

MT_I = moment about top joint I =

$$= -\left(V_A - P_1\right) XB_I + H_A YB_I + P_2\left(XB_I - XT_2\right) + \ \ldots \ + P_I\left(XB_I - XT_I\right)$$

and,

$$FT_{I,I+1} = \frac{MT_I}{d}$$

For a top joint I:

$$\Sigma M_I = 0$$

Yields,

MB_I = moment about bottom joint I =

$$= \left(V_A - P_1\right) XT_I - H_A YT_I - P_2\left(XT_I - XT_2\right) - \ \ldots \ - P_{I-1}\left(XT_I - XT_{I-1}\right)$$

and,

$$FB_{I-1,I} = \frac{MB_I}{d}$$

The shear and thrust values are calculated at the right of each joint. Refer to Figure 10 for the free body diagram. Positive values of shear and thrust are shown.



Figure 10. Free Body Diagrams for Shear and Thrust

From Figure 10:

$$V_I = \text{Shear in panel I} = \left(V_A - P_I - P_2 - \ldots - P_I\right) \sin\rho - H_A \cos\rho$$

$$THR_I = \text{Thrust in panel I} = \left(V_A - P_1 - P_2 - \ldots - P_I\right) \cos\rho + H_A \sin\rho$$



A. Section at Joint I B. Free Body of Joint I

Figure 11. Free Body Diagrams for Web Member Forces

Knowing V_I and T_I, the forces in the diagonal and radial web members can be found.

From the cut section in Figure 11,

$$FP_I = V_I \csc \theta$$

and from the free body of joint I in Figure 11,

$$FV_I = FP_I \sin\theta = V_I$$

where

$$\theta \cong \tan^{-1}\left[\frac{d}{\left(R - \frac{d}{2}\right) d\varphi}\right]$$

Also, it can be noted that the thrust is equal to the difference in the forces in the top and bottom chords.

CHAPTER III

COMPUTER PROGRAMS

The computer is very useful to solve problems such as the trial and error procedure required to find the location of the plastic members in the arch.

The procedure used in programing for the computer is the same as described in Chapter II. The input that is required for the computer is the span, the rise, the number of panels, and the values of the uniform loads. With this data, the computer will calculate all required geometry, the values of the concentrated loads, and the vertical reactions. Next, the joints K and L are selected and the value of the plastic moment is found. If the value of the plastic moment is negative, a new location for K and L is selected. If the plastic moment is positive, the horizontal reactions are determined and the moments across the arch are calculated. All positive moments are compared to the value of the plastic moment and if any moment exceeds the plastic value, a new location for joints K and L is tried. This cycle is repeated until all conditions are satisfied. Next, the values for shear and thrusts are found. The computer then punches the input data, the reactive forces, the plastic moment value, and the moment, shear and thrust values at points across the arch. If the conditions described above are not satisfied, the computer will punch the input data, and the statement that no plastic solution exists for the above data.

Program Number I is for live and dead loading only. The maximum number of panels, N, is fifty. Because of symmetry, the moment, shear, and thrust values are calculated for only one-half of the arch. From this, the member forces can be found for the entire arch.

Program Number II is for live, dead, and drift loading. It is programed so that the maximum number of panels, N, that can be read in is twenty-seven. This is the largest value that can be obtained on the IBM 1620 20K memory without overlap. If a computer with larger memory capacity were used, the only statements that would need to be changed to increase the allowable value of N would be the dimension statements. The program is used in this thesis in Chapter IV to illustrate the procedure to solve a numerical problem.

If the solution for live and dead loading only is desired, it is best to use Program Number I. Per data card, it takes approximately one-fifth the required computer time as Program Number II.

The computer programs are written in IBM 1620 Fortran (with Format) and compiled using the PDQ Fortran compiler for use in the example problem of Chapter IV.

The source programs are shown on pages 24 through 27. Flow diagrams of the computer programs are shown in Figures 12 and 13. The use of the programs is illustrated in Chapter

IV. The symbols used in the programs are defined under the List of Symbols.

PROGRAM NUMBER I

```
C     PLASTIC ANALYSIS OF TWO HINGED LATTICED CIRCULAR ARCHES
C     LIVE AND DEAD LOADING
      DIMENSION XB(27),XT(27),YB(27),YT(27),XMT(26),XMB(26),P(26)
      DIMENSION SH(26),THR(26)
100   FORMAT(2F10.4,I3,3F10.4)
101   FORMAT(4X,1HN,6X,4HSPAN,6X,4HRISE,5X,5HDEPTH,8X,2HWL,8X,2HWD)
102   FORMAT(I5,5F10.4//)
103   FORMAT(13X,2HVA,8X,2HVB,8X,2HHA,8X,2HHB,6X,14HPLASTIC MOMENT)
104   FORMAT(5X,4F10.4,F20.4//)
105   FORMAT(13X,6HMOMENT,7X,6HMOMENT)
106   FORMAT(45HJOINT       ABOUT TOP    ABOUT BOTTOM     PANEL,5X,
     115HSHEAR     THRUST)
107   FORMAT(I5,2F15.4,I10,2F10.4)
108   FORMAT(4X,45HNO PLASTIC SOLUTION EXISTS FOR THE ABOVE DATA)
5     READ 100,SP,T,N,D,WL,WD
      HS=SP/2.0
      R=((T**2)+(HS**2))/(2.0*T)
      E=R-T
      PHI=2.0*ATAN(HS/E)
      PN=N
      DPHI=PHI/PN
      ALFA=ATAN(E/HS)
      RHO=ALFA
      XT(1)=0.0
      K=(N+2)/2
      NK=K+1
      DO 10 I=2,NK
      RHO=RHO+DPHI
      XB(I)=HS-((R-(D/2.0))*COS(RHO))
      XT(I)=HS-((R+(D/2.0))*COS(RHO))
      YB(I)=((R-(D/2.0))*SIN(RHO))-E
10    YT(I)=((R+(D/2.0))*SIN(RHO))-E
      P(1)=(XT(2)*WL+WD*R*DPHI)/2.0
      DO 15 I=2,K
15    P(I)=(((XT(I+1)-XT(I-1))*WL)/2.0)+WD*R*DPHI
      VA=WL*HS+WD*R*PHI*0.5
      VB=VA
      L=1
      RA=VA-P(1)
20    L=L+1
      PM1=((RA*XT(K)*YB(L))-(RA*XB(L)*YT(K)))/(YB(L)+YT(K))
      PM2=0.0
      DO 25 I=2,L
25    PM2=PM2+((P(I)*(XB(L)-XT(I)))/(YB(L)+YT(K)))*YT(K)
      KN=K-1
      PM3=0.0
      DO 26 I=2,KN
26    PM3=PM3+P(I)*(XT(K)-XT(I))*YB(L)/(YB(L)+YT(K))
      PM=PM1+PM2-PM3
      IF (PM) 60,27,27
```

PROGRAM NUMBER I CONTINUED

```
27     HAN=PM+RA*XB(L)
       DO 30 I=2,L
30     HAN=HAN-(P(I)*(XB(L)-XT(I)))
       HA=HAN/YB(L)
       HB=-HA
       XMT(1)=0.0
       XMB(1)=0.0
       XMT(2)=(RA*XT(2))-(HA*YT(2))
       XMB(2)=-(RA*XB(2))+(HA*YB(2))+(P(2)*(XB(2)-XT(2)))
       DO 51 I=3,K
       XMT(I)=(RA*XT(I))-(HA*YT(I))
       XMB(I)=-(RA*XB(I))+(HA*YB(I))
       NI=I-1
       DO 40 M=2,NI
40     XMT(I)=XMT(I)-(P(M)*(XT(I)-XT(M)))
       DO 50 M=2,I
50     XMB(I)=XMB(I)+(P(M)*(XB(I)-XT(M)))
51     CONTINUE
       I=1
52     I=I+1
       IF (XMB(I)) 54,54,53
53     DIF2=XMB(I)-PM-0.01
       IF (DIF2) 54,54,60
54     IF (I-K+1) 52,55,55
55     RVA=VA
       RHO=ALFA-DPHI
       DO 57 I=1,K
       RVA=RVA-P(I)
       RHO=RHO+DPHI
       SH(I)=RVA*SIN(RHO)-HA*COS(RHO)
57     THR(I)=RVA*COS(RHO)+HA*SIN(RHO)
       PUNCH 101
       PUNCH 102,N,SP,T,D,WL,WD
       PUNCH 103
       PUNCH 104,VA,VB,HA,HB,PM
       PUNCH 105
       PUNCH 106
       DO 59 I=1,K
       PUNCH 107,I,XMT(I),XMB(I),I,SH(I),THR(I)
59     CONTINUE
       GO TO 5
60     IF (L-K+1) 20,65,65
65     PUNCH 101
       PUNCH 102,N,SP,T,D,WL,WD
       PUNCH 108
       GO TO 5
       END
```

PROGRAM NUMBER II

```
C     PLASTIC ANALYSIS OF TWO HINGED LATTICED CIRCULAR ARCHES
C     LIVE,DEAD,AND DRIFT LOADING
      DIMENSION XB(26),XT(28),YB(26),YT(26),XMT(28),XMB(28),P(28)
      DIMENSION SH(28),THR(28)
100   FORMAT(2F10.4,I3,4F10.4)
101   FORMAT(4X,1HN,6X,4HSPAN,6X,4HRISE,5X,5HDEPTH,8X,2HWL,8X,2HWD,7X,
     13HWDR)
102   FORMAT(I5,6F10.4//)
103   FORMAT(13X,2HVA,8X,2HVB,8X,2HHA,8X,2HHB,6X,14HPLASTIC MOMENT)
104   FORMAT(5X,4F10.4,F20.4//)
105   FORMAT(13X,6HMOMENT,7X,6HMOMENT)
106   FORMAT(45HJOINT       ABOUT TOP   ABOUT BOTTOM      PANEL,5X,
     115HSHEAR     THRUST)
107   FORMAT(I5,2F15.4,I10,2F10.4)
108   FORMAT(4X,45HNO PLASTIC SOLUTION EXISTS FOR THE ABOVE DATA)
5     READ 100,SP,T,N,D,WL,WD,WDR
      HS=SP/2.0
      R=((T**2)+(HS**2))/(2.0*T)
      E=R-T
      PHI=2.0*ATAN(HS/E)
      PN=N
      DPHI=PHI/PN
      ALFA=ATAN(E/HS)
      RHO=ALFA
      XT(1)=0.0
      NN=N+1
      XT(NN)=SP
      DO 10 I=2,N
      RHO=RHO+DPHI
      XB(I)=HS-((R-(D/2.0))*COS(RHO))
      XT(I)=HS-((R+(D/2.0))*COS(RHO))
      YB(I)=((R-(D/2.0))*SIN(RHO))-E
10    YT(I)=((R+(D/2.0))*SIN(RHO))-E
      P(1)=(XT(2)*WL+WD*R*DPHI)/2.0
      KCL=(N+2)/2
      KCN=KCL-1
      KCP=KCL+1
      DO 15 I=2,KCN
15    P(I)=(((XT(I+1)-XT(I-1))*WL)/2.0)+WD*R*DPHI
      P(KCL)=(((XT(KCP)-XT(KCN))*WL+(XT(KCP)-XT(KCL))*WDR)/2.)+WD*R*DPHI
      DO 16 I=KCP,N
16    P(I)=(((XT(I+1)-XT(I-1))*(WL+WDR))/2.0)+WD*R*DPHI
      P(NN)=(XT(2)*(WL+WDR)+WD*R*DPHI)/2.0
      VA=WL*HS+WD*R*PHI*0.5+WDR*HS*0.25
      VB=WL*HS+WD*R*PHI*0.5+WDR*HS*0.75
      RA=VA-P(1)
      K=N
18    L=1
      K=K-1
20    L=L+1
      PM1=((RA*XT(K)*YB(L))-(RA*XB(L)*YT(K)))/(YB(L)+YT(K))
      PM2=0.0
      DO 25 I=2,L
25    PM2=PM2+((P(I)*(XB(L)-XT(I)))/(YB(L)+YT(K)))*YT(K)
      KN=K-1
      PM3=0.0
      DO 26 I=2,KN
26    PM3=PM3+P(I)*(XT(K)-XT(I))*YB(L)/(YB(L)+YT(K))
      PM=PM1+PM2-PM3
      IF (PM) 60,27,27
```

PROGRAM NUMBER II CONTINUED

```
27    HAN=PM+RA*XB(L)
      DO 30 I=2,L
30    HAN=HAN-(P(I)*(XB(L)-XT(I)))
      HA=HAN/YB(L)
      HB=-HA
      XMT(1)=0.0
      XMB(1)=0.0
      XMT(2)=(RA*XT(2))-(HA*YT(2))
      XMT(NN)=0.0
      XMB(NN)=0.0
      XMB(2)=-(RA*XB(2))+(HA*YB(2))+(P(2)*(XB(2)-XT(2)))
      DO 40 I=3,N
      XMT(I)=(RA*XT(I))-(HA*YT(I))
      XMB(I)=-(RA*XB(I))+(HA*YB(I))
      NI=I-1
      DO 35 M=2,NI
35    XMT(I)=XMT(I)-(P(M)*(XT(I)-XT(M)))
      DO 40 M=2,I
40    XMB(I)=XMB(I)+(P(M)*(XB(I)-XT(M)))
      I=1
42    I=I+1
      IF (XMT(I)) 45,45,44
44    DIF1=XMT(I)-PM-0.01
      IF (DIF1) 45,45,60
45    IF (XMB(I)) 48,48,46
46    DIF2=XMB(I)-PM-0.01
      IF (DIF2) 48,48,60
48    IF (I-N) 42,50,50
50    RVA=VA
      RHO=ALFA-DPHI
      DO 55 I=1,N
      RVA=RVA-P(I)
      RHO=RHO+DPHI
      SH(I)=RVA*SIN(RHO)-HA*COS(RHO)
55    THR(I)=RVA*COS(RHO)+HA*SIN(RHO)
      RHO=RHO+DPHI
      SH(NN)=(RVA-P(NN))*SIN(RHO)-HA*COS(RHO)
      THR(NN)=(RVA-P(NN))*COS(RHO)+HA*SIN(RHO)
      PUNCH 101
      PUNCH 102,N,SP,T,D,WL,WD,WDR
      PUNCH 103
      PUNCH 104,VA,VB,HA,HB,PM
      PUNCH 105
      PUNCH 106
      DO 56 I=1,NN
      PUNCH 107,I,XMT(I),XMB(I),I,SH(I),THR(I)
56    CONTINUE
      GO TO 5
60    IF (L-KCP) 20,65,65
65    IF (K-KCN) 70,70,18
70    PUNCH 101
      PUNCH 102,N,SP,T,D,WL,WD,WDR
      PUNCH 108
      GO TO 5
      END
```



Figure 12. Flow Diagram for Program Number I



Figure 13. Flow Diagram for Program Number II

CHAPTER IV

EXAMPLE PROBLEM

An example arch will be studied for the following data:

Arches at 12'-0" o. c.

Span = 100'-0"

Rise = 25'-0"

Loading:

Dead Load = 200 lbs./ft.

Live Load = 360 lbs./ft.

Drift Load = 180 lbs./ft.

Combinations of Loads:

I. 1.85 (W_L+W_D)

II. 1.85 $(W_{DR}+W_D)$

III. 1.85 $(2W_{DR}+W_D)$

Number of Panels = N = 20

Depth = Varies

For load combination Number I, Computer Program Number I is used. An example data card for this case is shown in Figure 14. PDQ Free Form Subroutines are used so that the input data cards are punched with one space between each entry.

The data must be in the following order: The span, rise, number of panels, depth, live load, and dead load. All of the values except the number of joints, N, must contain a decimal. N must be a fixed point number.

Using a depth of zero, the following values are obtained:

N	SPAN	RISE	DEPTH	WL	WD
20	100.0000	25.0000	.0000	.6700	.3700

VA	VB	HA	HB	PLASTIC MOMENT
54.9437	54.9437	51.2759	-51.2759	50.0749

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-9.4389	72.8746
2	-35.0580	35.0580	2	-5.8388	69.9171
3	-50.0749	50.0749	3	-3.0279	66.7772
4	-49.6679	49.6679	4	-1.0005	63.6073
5	-38.3787	38.3787	5	.2765	60.5553
6	-20.5179	20.5179	6	.8626	57.7596
7	-.0160	.0160	7	.8406	55.3444
8	19.7102	-19.7102	8	.3147	53.4160
9	35.8752	-35.8752	9	-.5936	52.0592
10	46.4183	-46.4183	10	-1.7499	51.3344
11	50.0750	-50.0750	11	-3.0109	51.2759

From the above data, the distance to the pressure line can be approximated.

$$d_p = \frac{PM}{T} = \frac{50.0749}{66.7772} = 0.7499 \text{ ft.}$$

Therefore, for a depth of the arch equal to twice d_p, which is equal to 1.4998 ft., the plastic moment should be equal to zero and there will be no tension in the chord members. For depths greater than this value, there should be no plastic solution for the arch. To find this depth, a trial and error procedure is followed. The depth is varied and the following answers are obtained. Only the last two trials are shown. The depth of 1.690 ft. found below compares to twice the estimated depth to the pressure line 1.4998 ft.

N	SPAN	RISE	DEPTH	WL	WD
20	100.0000	25.0000	1.6900	.6700	.3700

VA	VB	HA	HB	PLASTIC MOMENT
54.9437	54.9437	51.2626	-51.2626	.0041

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-9.3025	73.0344
2	-96.4021	-22.2451	2	-5.5597	70.2054
3	-109.9063	-3.3443	3	-2.7518	67.0122
4	-107.8102	.0041	4	-.7371	63.7905
5	-94.7733	-7.7924	5	.5180	60.6898
6	-75.2180	-22.5486	6	1.0735	57.8502
7	-53.1771	-40.4444	7	1.0135	55.3973
8	-32.1556	-58.1562	8	.4433	53.4389
9	-15.0115	-72.9708	9	-.5141	52.0606
10	-3.8590	-82.8780	10	-1.7226	51.3236
11	.0043	-86.6382	11	-3.0371	51.2626

N	SPAN	RISE	DEPTH	WL	WD
20	100.0000	25.0000	1.6910	.6700	.3700

NO PLASTIC SOLUTION EXISTS FOR THE ABOVE DATA

Usual depth to span ratios for latticed arches are 1/150 and 1/100. For a ratio of 1/150 for this example will give a depth of the arch equal to 0.6667 ft. and for a ratio of 1/100, the depth will be 1.0 ft. Answers for these depths are shown on the following page. It can be noted that for a ratio of 1/150, the maximum force in the compressive member is 2.4410 times as great as the maximum tensile force. For a span/depth of 1/100, the maximum compressive force is 4.1961 times as great as the maximum tensile force. Also,

N	SPAN	RISE	DEPTH	WL	WD
20	100.0000	25.0000	.6667	.6700	.3700

VA	VB	HA	HB	PLASTIC MOMENT
54.9437	54.9437	51.2757	-51.2757	30.1916

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-9.3891	72.9407
2	-59.2521	12.5602	2	-5.7324	70.0342
3	-73.6987	29.1140	3	-2.9224	66.8736
4	-72.6495	30.1916	4	-.8997	63.6836
5	-60.6927	20.2821	5	.3691	60.6127
6	-42.1824	3.6471	6	.9435	57.7998
7	-21.0892	-15.8258	7	.9070	55.3700
8	-.8646	-34.7571	8	.3641	53.4299
9	15.6773	-50.3889	9	-.5632	52.0647
10	26.4554	-60.6806	10	-1.7396	51.3352
11	30.1917	-64.3773	11	-3.0212	51.2757

N	SPAN	RISE	DEPTH	WL	WD
20	100.0000	25.0000	1.0000	.6700	.3700

VA	VB	HA	HB	PLASTIC MOMENT
54.9437	54.9437	51.2709	-51.2709	20.3712

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-9.3605	72.9709
2	-71.3401	1.2505	2	-5.6758	70.0896
3	-85.4796	18.5611	3	-2.8665	66.9184
4	-84.0892	20.3711	4	-.8464	63.7180
5	-71.7806	11.1432	5	.4179	60.6374
6	-52.9301	-4.8857	6	.9861	57.8157
7	-31.5283	-23.8500	7	.9419	55.3784
8	-11.0443	-42.3879	8	.3900	53.4323
9	5.6936	-57.7565	9	-.5471	52.0628
10	16.5934	-67.9244	10	-1.7340	51.3309
11	20.3712	-71.6422	11	-3.0264	51.2709

the compression member must be designed so that buckling will not occur. Therefore, the compression chord member will be much larger than the tension chord member.

Computer Program Number II is used for load combination number II. An example data card is shown in Figure 14. The cards are punched in the following order: Span, rise, number of panels, depth, live load, dead load, and drift load. All punched values must contain a decimal except the number of panels, N, and it must be a fixed point value.



Figure 14. Example Data Cards

For a depth of zero the following answers are obtained:

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	.0000	.0000	.3700	.3300

VA	VB	HA	HB	PLASTIC MOMENT
25.5687	33.8187	26.8730	-26.8730	60.7752

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-6.8005	35.7210
2	-29.7647	29.7647	2	-4.9035	34.6082
3	-48.8502	48.8502	3	-3.2588	33.4648
4	-58.7240	58.7240	4	-1.8543	32.3265
5	-60.7752	60.7752	5	-.6739	31.2269
6	-56.2907	56.2907	6	.3024	30.1961
7	-46.4320	46.4320	7	1.0986	29.2611
8	-32.2165	32.2165	8	1.7415	28.4448
9	-14.4999	14.4999	9	2.2603	27.7659
10	6.0369	-6.0369	10	2.6865	27.2388
11	28.8987	-28.8987	11	2.0979	26.8730
12	48.2558	-48.2558	12	.5486	26.9379
13	58.6643	-58.6643	13	-.9117	27.5129
14	60.7752	-60.7752	14	-2.1879	28.5723
15	55.7849	-55.7849	15	-3.1903	30.0746
16	45.3973	-45.3973	16	-3.8376	31.9637
17	31.7708	-31.7708	17	-4.0594	34.1707
18	17.4529	-17.4529	18	-3.7981	36.6156
19	5.3035	-5.3035	19	-3.0102	39.2101
20	-1.5890	1.5890	20	-1.6677	41.8597
21	.0000	.0000	21	1.2071	43.1787

The approximate depth to the pressure line,

$$d_p = \frac{PM}{T} = \frac{60.7752 \text{ ft. kips}}{31.2269 \text{ kips}} = 1.9463 \text{ ft.}$$

The last two trials to find the critical depth are shown on the following page.

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	4.1510	.0000	.3700	.3300

VA	VB	HA	HB	PLASTIC MOMENT
25.5687	33.8187	26.3438	-26.3438	.0009

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-6.3771	35.4035
2	-101.7898	-40.3940	2	-4.5114	34.2529
3	-119.2630	-18.0299	3	-2.9012	33.0746
4	-127.4349	-5.0027	4	-1.5344	31.9050
5	-127.7585	.0009	5	-.3944	30.7775
6	-121.5786	-1.8009	6	.5391	29.7228
7	-110.1089	-9.3069	7	1.2904	28.7679
8	-94.4111	-21.5511	8	1.8868	27.9360
9	-75.3784	-37.7191	9	2.3579	27.2458
10	-53.7203	-57.1606	10	2.7355	26.7119
11	-29.9526	-79.4006	11	2.0661	26.3438
12	-10.2892	-99.3795	12	.4052	26.4198
13	-.5364	-111.6312	13	-1.1638	27.0218
14	.0009	-116.7414	14	-2.5430	28.1234
15	-7.4028	-115.8069	15	-3.6405	29.6819
16	-20.9366	-110.4002	16	-4.3730	31.6398
17	-38.3069	-102.5205	17	-4.6683	33.9261
18	-56.8087	-94.5329	18	-4.4673	36.4591
19	-73.4058	-89.0969	19	-3.7254	39.1478
20	-84.8213	-88.2439	20	-2.2302	41.6924
21	.0000	.0000	21	.7838	42.8612

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	4.1511	.0000	.3700	.3300

NO PLASTIC SOLUTION EXISTS FOR THE ABOVE DATA

The critical depth of 4.1510 ft. compares to twice the estimated depth to the pressure line of 3.8926 ft.

The moments, shears and thrusts for depth to span ratios of 1/150 and 1/100 are shown on the following pages. For a depth/span ratio of 1/150, the largest compressive force is 1.4082 times as large as the maximum tensile force. For a depth/span ratio of 1/100, the largest compressive force is

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	.6667	.0000	.3700	.3300

VA	VB	HA	HB	PLASTIC MOMENT
25.5687	33.8187	26.7877	-26.7877	50.8846

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-6.7323	35.6698
2	-41.4311	18.3959	2	-4.8404	34.5510
3	-60.2664	37.9973	3	-3.2012	33.4019
4	-69.8747	48.3678	4	-1.8028	32.2586
5	-71.6554	50.8846	5	-.6289	31.1545
6	-66.9045	46.8235	6	.3405	30.1199
7	-56.7921	37.3367	7	1.1295	29.1817
8	-42.3424	23.4329	8	1.7649	28.3628
9	-24.4172	5.9615	9	2.2760	27.6821
10	-3.7020	-14.4014	10	2.6943	27.1539
11	19.3046	-37.1640	11	2.0928	26.7877
12	38.7123	-56.6162	12	.5255	26.8544
13	49.0196	-67.3098	13	-.9523	27.4338
14	50.8846	-69.8857	14	-2.2450	28.5000
15	45.5160	-65.5246	15	-3.2627	30.0113
16	34.6347	-55.9100	16	-3.9237	31.9115
17	20.4203	-43.1757	17	-4.1574	34.1312
18	5.4458	-29.8405	18	-3.9058	36.5903
19	-7.4011	-18.7332	19	-3.1253	39.1999
20	-15.0028	-12.8871	20	-1.7582	41.8327
21	.0000	.0000	21	1.1389	43.1276

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	1.0000	.0000	.3700	.3300

VA	VB	HA	HB	PLASTIC MOMENT
25.5687	33.8187	26.7452	-26.7452	45.9587

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-6.6982	35.6443
2	-47.2493	12.7268	2	-4.8088	34.5224
3	-65.9582	32.5876	3	-3.1724	33.3705
4	-75.4326	43.2078	4	-1.7770	32.2247
5	-77.0771	45.9587	5	-.6064	31.1183
6	-72.1922	42.1104	6	.3596	30.0818
7	-61.9522	32.8102	7	1.1450	29.1420
8	-47.3849	19.0630	8	1.7766	28.3219
9	-29.3550	1.7147	9	2.2839	27.6403
10	-8.5504	-18.5611	10	2.6983	27.1115
11	14.5287	-41.2739	11	2.0902	26.7452
12	33.9615	-60.7744	12	.5140	26.8128
13	44.2177	-71.6120	13	-.9725	27.3943
14	45.9588	-74.4227	14	-2.2736	28.4638
15	40.3997	-70.3795	15	-3.2988	29.9797
16	29.2698	-61.1552	16	-3.9667	31.8854
17	14.7597	-48.8713	17	-4.2063	34.1115
18	-.5455	-36.0322	18	-3.9595	36.5776
19	-13.7437	-25.4512	19	-3.1827	39.1949
20	-21.7018	-20.1171	20	-1.8034	41.8192
21	.0000	.0000	21	1.1049	43.1020

1.6771 times as large as the maximum tensile force.

Comparing load case I to load case II, it is found that for depths greater than zero, case II produced larger tensile forces in the chord members and case I yields the larger compressive force values. Therefore, the arch must be designed considering both load cases.

If it is possible that the drift load may be applied to both halves of the arch, plastic design will not yield a savings of material because almost all of the members must be designed as compressive members.

Now consider the effect of increasing the drift load to a value equal to the live load as in load case III. The answers for a depth/span ratio of 1/100 are shown on the following page.

N	SPAN	RISE	DEPTH	WL	WD	WDR
20	100.0000	25.0000	1.0000	.0000	.3700	.6700

VA	VB	HA	HB	PLASTIC MOMENT
29.8187	46.5687	34.8151	-34.8151	94.8661

JOINT	MOMENT ABOUT TOP	MOMENT ABOUT BOTTOM	PANEL	SHEAR	THRUST
1	.0000	.0000	1	-10.6041	43.8862
2	-71.9250	28.8344	2	-7.9348	43.0906
3	-106.5505	64.3591	3	-5.4916	42.1914
4	-127.1665	85.9440	4	-3.2694	41.2224
5	-135.0817	94.8661	5	-1.2593	40.2156
6	-131.5429	92.3425	6	.5519	39.2004
7	-117.7128	79.5092	7	2.1808	38.2036
8	-94.6500	57.4013	8	3.6471	37.2487
9	-63.2923	26.9367	9	4.9729	36.3555
10	-24.4420	-11.0984	10	6.1828	35.5403
11	21.2456	-56.0607	11	5.3485	34.8151
12	61.8703	-96.5995	12	2.5391	34.7292
13	86.0830	-121.5316	13	-.1391	35.4486
14	94.8661	-131.7964	14	-2.5317	36.9301
15	90.0927	-129.1963	15	-4.4948	39.1036
16	74.4559	-116.3301	16	-5.8994	41.8742
17	51.3751	-96.5010	17	-6.6360	45.1258
18	24.8773	-73.6020	18	-6.6176	48.7247
19	-.5389	-51.9848	19	-5.7826	52.5238
20	-20.0560	-36.2119	20	-4.0066	56.2681
21	.0000	.0000	21	-.0891	58.1440

Load case III gives larger compressive and tensile forces than the live and dead load condition of load case I. This illustrates that the relative magnitudes of the loads affects the load case that must be used for design.

CHAPTER V

SUMMARY AND CONCLUSIONS

Plastic design can be extended to the design of latticed arches if the compression members are designed so that they will not buckle until the structure as a whole fails. The force in the members may change from compression to tension with different types of loads. Therefore, the designer must be sure that he considers all of the different loads that may be imposed upon the structure. This will call for good engineering judgment.

The trial and error procedure required for the solution would not be easy by hand unless the position of the plastic members could be predicted in advance. But with the aid of the computer, the design is simplified. The span, rise, number of panels, and loading is all that must be determined in advance. When the computer answers are obtained, it is a simple procedure to find the member forces and determine the required sizes.

It was found that for depths greater than zero, the forces in the compressive chord members are greater than the forces in the tension chord members. The compression chord required will therefore be much larger than the tension chord. The difference is dependent upon the pitch of the arch, whether the load is symmetrical or unsymmetrical, and the relative magnitudes of the loads.

Suggestions for Future Study

The accuracy of a theoretical solution depends on the degree of validity of presumptions. Therefore, testing of structures should be done to determine if the mode of collapse and failure load is the same as predicted by the theoretical solution.

Other arches such as hingeless circular arches, circular arches with different support elevations, parabolic arches, arches with variable cross section, and catenaries should be investigated to see if the theory of plastic design could be applied to these types of structures.

A SELECTED BIBLIOGRAPHY

1. Beedle, Lynn S. Plastic Design of Steel Frames. New York: John Wiley and Sons, Inc., 1958.

2. Beedle, Lynn S. et al., Structural Steel Design. New York: The Ronald Press Co., 1964.

3. Cornforth, Robert C. "Plastic Analysis of Two Hinged Circular Arches." (Unpublished Thesis, Oklahoma State University, 1964.).

4. Fisher, B. H., Heyman, Jacques, L. G. "The Plastic Design of Latticed Portal Frames." The Structural Engineer. October, 1961, 318-331.

5. Neal, B. G. The Plastic Methods of Structural Analysis. New York: John Wiley and Sons, Inc., 1963.

6. Van Den Broek, J. A. Theory of Limit Design. New York: John Wiley and Sons, Inc., 1948, 62-126.

VITA

Mervin L. Snowden

Candidate for the Degree of

Master of Architectural Engineering

Thesis: PLASTIC DESIGN OF TWO HINGED LATTICED CIRCULAR ARCHES

Major Field: Architectural Engineering

Biographical:

Personal Data: Born in Enid, Oklahoma, November 19, 1941, the son of Harold L. and Opal I. Snowden.

Education: Attended grade school in Billings, and Enid, Oklahoma; graduated from Enid High School, Enid, Oklahoma, in 1959; received the degree of Bachelor of Architectural Engineering from Oklahoma State University in May, 1964; completed requirements for the Master of Architectural Engineering degree in May, 1966.

Professional experience: Sullivan Engineering Company, Oklahoma City, Oklahoma, from May, 1964 to September, 1964, and from May, 1965 to September, 1965. Graduate Assistant, School of Architecture, Oklahoma State University, from September, 1964 to January, 1966.

Organizations: Sigma Tau, Chi Epsilon, Scabbard and Blade, Phi Kappa Phi.